A Conjecture on Primitive Pythagorean Triples

R. Scott McIntire

Aug 11, 2026

1 Overview

Some numbers carry a right triangle on their backs and some do not. The number 5 is the hypotenuse of the right triangle with sides $(3, 4, 5)$; the numbers 7 and 11 are hypotenuses of no integer right triangle at all. The number 25 is a hypotenuse in exactly one primitive way, $(7, 24, 25)$, while 65 manages it twice: $(16, 63, 65)$ and $(33, 56, 65)$. Two questions suggest themselves: *which* numbers occur as hypotenuses of primitive Pythagorean triples, and *in how many ways*? Classical number theory answers both questions completely – and the answers set up a third question, about the primes among these hypotenuses, that appears to be genuinely open.

In this note we discuss the set $\mathcal{H}$ of primitive Pythagorean hypotenuses – the "hypotenuse" values that arise from primitive Pythagorean triples (right triangles with integer sides sharing no common factor) – recall the classical theorems describing its structure, and state an open conjecture about the cumulative parity of its prime elements.

A related set, $\mathcal{H}_p \subset \mathcal{H}$, consists of those elements of $\mathcal{H}$ that are prime – the primitive Pythagorean hypotenuses that are themselves prime numbers.

Two of the three observations below turn out to be classical theorems – we recall them for orientation – while the third remains an open conjecture of mine, which is the focus of this note.

1. **Partition of $\mathcal{H}$** (theorem): $\mathcal{H}$ is *partitioned* into two disjoint infinite sets: the set of all powers of Pythagorean primes, and the set of all hypotenuses arising from more than one primitive Pythagorean triple.
2. **Infinitude of $\mathcal{H}_p$** (Dirichlet): The set of Pythagorean primes is infinite.
3. $\mathcal{H}_p \sim \mathbb{P}$ (conjecture): $\mathcal{H}_p$ is *similar* to and as *complex* as the primes when measured by certain cumulative-parity metrics.

2 Definitions

After a few definitions we describe a conjecture regarding Primitive Pythagorean Triples. We use standard notation, associating the natural numbers with the symbol $\mathbb{N}$[1] and we use the symbol $\mathbb{P}$ to denote the set of prime numbers.

Definition 1. *Pythagorean Triples*, denoted $\mathcal{T}$, are triples of numbers representing the lengths of the sides of right triangles where all the sides are integers.[2] Specifically,

$$\mathcal{T} = \{(x, y, z) \mid (x, y, z) \in \mathbb{N}^3 \ \wedge \ x^2 + y^2 = z^2\}. \tag{1}$$

Definition 2. *Primitive Pythagorean Triples*, denoted $\mathcal{T}_p$, are Pythagorean triples whose legs share no common factor:[3]

$$\mathcal{T}_p = \{(x, y, z) \mid (x, y, z) \in \mathbb{N}^3 \ \wedge \ \gcd(x, y) = 1 \ \wedge \ x^2 + y^2 = z^2\}. \tag{2}$$

Definition 3. *Primitive Pythagorean Hypotenuses*, denoted $\mathcal{H}$, are defined by:

$$\mathcal{H} = \{z \mid \exists (x, y) \in \mathbb{N}^2, \ (x, y, z) \in \mathcal{T}_p\}. \tag{3}$$

This is the set of all possible hypotenuses of primitive Pythagorean triples.

Definition 4. *Pythagorean Primes*, denoted $\mathcal{H}_p$, are defined by:

$$\mathcal{H}_p = \{z \mid \exists (x, y) \in \mathbb{N}^2, (x, y, z) \in \mathcal{T}_p \ \wedge \ z \in \mathbb{P}\}. \tag{4}$$

This is the set of primes that occur as hypotenuses of primitive Pythagorean triples.

Definition 5. *Duplicate Primitive Hypotenuses*, denoted $\mathcal{H}_d$, are defined by:

$$\mathcal{H}_d = \{z \mid \exists \, (x_1, x_2, y_1, y_2) \in \mathbb{N}^4, \ (x_1, y_1, z) \in \mathcal{T}_p \ \wedge \ (x_2, y_2, z) \in \mathcal{T}_p \ \wedge \ \{x_1, y_1\} \neq \{x_2, y_2\}\}. \tag{5}$$

This is the set of primitive hypotenuses that arise from more than one *distinct* primitive Pythagorean triple (i.e. not merely a swap of legs).

Definition 6. *Primitive Hypotenuses Powers*, denoted $\mathcal{H}_u$, are defined by:

$$\mathcal{H}_u = \{z \mid \exists \, (p, n) \in \mathcal{H}_p \times \mathbb{N}, \ z = p^n\}. \tag{6}$$

This is the set of all powers of Pythagorean primes.

Notes:

- The subscript u in the name $\mathcal{H}_u$ stands for *unique*: by Theorem 3 below, the elements of $\mathcal{H}_u$ are exactly the hypotenuses arising from a unique primitive Pythagorean triple.

[1] We assume that $\mathbb{N}$ does *not* include 0.

[2] Triples are taken as ordered, so $(3, 4, 5)$ and $(4, 3, 5)$ are distinct elements of $\mathcal{T}$.

[3] Since any common divisor of x and y also divides z, $\gcd(x, y) = 1$ is equivalent to $\gcd(x, y, z) = 1$, the usual primality condition.

- $\mathcal{H}_p$ is a *proper* subset of the primes: $\mathcal{H}_p \subset \mathbb{P}$.
- If the sets above are treated as lists, they are indexed in sorted order.

The ground floor of this discussion is the parity structure of a primitive triple: the hypotenuse is odd, one leg is even, and the even leg is divisible by 4. These facts follow from elementary parity arguments; an alternative *geometric* proof – via the inscribed circle of the right triangle, which turns out to have integer radius – appears in the companion note [4]. The present paper starts where that one stops: knowing the shape of a primitive triple, we ask which numbers occur as its hypotenuse.

First we list a few facts and state a couple of known theorems before proceeding:

1. The Pythagorean primes are exactly the primes of the form $4n + 1$. The key input is a theorem of Pierre de Fermat, known as Fermat's theorem on sums of two squares [1]. According to this theorem, an odd prime number p can be expressed as the sum of two squares if and only if p is congruent to 1 modulo 4, i.e., $p = 4n+1$ for some positive integer n. To pass from sums of two squares to hypotenuses, note that if $p = a^2 + b^2$ with $a > b$, then $p^2 = (a^2 - b^2)^2 + (2ab)^2$ is a primitive representation, so $p \in \mathcal{H}_p$; conversely, a classical argument in the Gaussian integers $\mathbb{Z}[i]$ shows that every prime factor of a primitive hypotenuse is congruent to 1 modulo 4.
2. Moreover, the distribution of primes of the form $4n + 1$ has been studied extensively. The German mathematician Peter Gustav Lejeune Dirichlet proved that there are infinitely many primes of the form $4n+1$, a result known as Dirichlet's theorem on arithmetic progressions [2]. This theorem, which generalizes the case of Pythagorean primes, states that for any two positive coprime numbers a and d, there are infinitely many primes of the form $a + nd$, where n is a non-negative integer.
3. In the case of Pythagorean primes, we can take $a = 1$ and $d = 4$ to get the form $4n + 1$. Dirichlet's theorem thus guarantees that there are infinitely many Pythagorean primes. This is a profound result, as it extends Euclid's theorem on the infinitude of the primes to arithmetic progressions.

From these results we get the following characterization of Pythagorean primes:

Theorem 1. $\mathcal{H}_p = \{p \mid p \in \mathbb{P} \ \wedge \ p \equiv 1 \pmod 4\}$.

And the following structure theorem for Pythagorean hypotenuses:

Theorem 2. *A natural number $z > 1$ belongs to $\mathcal{H}$ if and only if every prime factor of z lies in $\mathcal{H}_p$. Consequently, every $h \in \mathcal{H}$ factors uniquely over $\mathcal{H}_p$: there are a unique $N \in \mathbb{N}$, unique $p_1 < p_2 < \cdots < p_N$ in $\mathcal{H}_p$, and unique $(k_1, \dots, k_N) \in \mathbb{N}^N$ such that*

$$h = \prod_{i=1}^{N} p_i^{k_i}. \tag{7}$$

A classical consequence is the following partition theorem, which I had originally stated as a conjecture but which follows from the count of primitive representations of n as a sum of two squares:

Theorem 3. *The set $\mathcal{H}$ is partitioned by $\mathcal{H}_d$ and $\mathcal{H}_u$:*

(a) $\mathcal{H} = \mathcal{H}_d \cup \mathcal{H}_u$*;*

(b) $\emptyset = \mathcal{H}_d \cap \mathcal{H}_u$*;*

(c) $|\mathcal{H}_d| = \infty$*.*

Proof sketch. For $n \in \mathcal{H}$, the number of primitive Pythagorean triples with hypotenuse n equals 2^{k-1}, where k is the number of distinct prime factors of n (all of which lie in $\mathcal{H}_p$ by Theorem 2). This count is a classical consequence of the unique factorization of n in the Gaussian integers [3]. Hence n has exactly one primitive representation iff $k = 1$, i.e. iff n is a power of a single Pythagorean prime ($n \in \mathcal{H}_u$); otherwise $n \in \mathcal{H}_d$. This gives the partition. The infinitude of $\mathcal{H}_d$ follows from the infinitely many distinct products $p \cdot q$ with $p, q \in \mathcal{H}_p$ and $p \neq q$ (Dirichlet). □

Remark 1. The set $\mathcal{H}_u$ is also infinite – it contains $\mathcal{H}_p$, which is infinite by Dirichlet's theorem – so $\mathcal{H}$ is indeed partitioned into two disjoint *infinite* sets, as described in the overview.

It is worth naming the single strategic move behind all of the classical results quoted above: *change the representation.* In the Gaussian integers $\mathbb{Z}[i]$, the statement "z is a primitive Pythagorean hypotenuse" becomes "$z^2 = (x + iy)(x - iy)$ with x, y coprime" – a statement about *factorization* – and questions about triples become bookkeeping about how z factors: Fermat's theorem classifies which primes survive the change of representation, and the 2^{k-1} count records how many essentially distinct ways the prime factors of z can be split between $x + iy$ and its conjugate. Once the representation is right, the theorems are inventory.

The following table shows the inventory for the smallest members of each family, with the number of primitive triples doubling at each new distinct prime factor (each row verified by direct computation):

distinct primes k	triples 2^{k-1}	smallest example	lies in
1	1	5: $(3, 4, 5)$	$\mathcal{H}_u$
1	1	$25 = 5^2$: $(7, 24, 25)$	$\mathcal{H}_u$
2	2	$65 = 5 \cdot 13$: $(16, 63, 65)$, $(33, 56, 65)$	$\mathcal{H}_d$
3	4	$1105 = 5 \cdot 13 \cdot 17$: $(47, 1104, 1105), \ldots$	$\mathcal{H}_d$
4	8	$32045 = 5 \cdot 13 \cdot 17 \cdot 29$: $(716, 32037, 32045), \ldots$	$\mathcal{H}_d$

Theorem 3 is this table read down its last column: the $k = 1$ rows – the prime powers – are exactly $\mathcal{H}_u$, and every other row lands in $\mathcal{H}_d$.

3 McIntire's Pythagorean Triple Conjecture

Having settled which numbers are hypotenuses and in how many ways, we turn to the open question: how do the *prime* hypotenuses sit inside the primes? Both $\mathcal{H}_p$ and $\mathbb{P}$ are infinite sorted lists, and every prime $p \geq 5$ falls into one of two camps – $p \equiv 1$ or $p \equiv -1 \pmod 6$. Assign each camp a value:

Definition 7. The *Prime Parity* function, Π, is defined on any prime p with $p \geq 5$ by:

$$\Pi(p) = \begin{cases} 1 & \exists n \in \mathbb{N},\ p = 6n+1 \\ -1 & \exists n \in \mathbb{N},\ p = 6n-1 \end{cases} \tag{8}$$

Now run a race between two walkers on the integers, both starting at 0. The first walker steps through the Pythagorean primes $5, 13, 17, 29, 37, \ldots$ in order, moving up or down by Π at each step; the second steps through *all* primes ≥ 5: that is, $5, 7, 11, 13, 17, \ldots$. After m steps the walkers stand at the *cumulative parities* $\sum_{i=1}^{m} \Pi(\mathcal{H}_p[i])$ and $\sum_{i=3}^{m+2} \Pi(\mathbb{P}[i])$ respectively (the second sum starts at index 3 so as to skip the primes 2 and 3, on which Π is undefined; both sums thus range over primes ≥ 5). The first few steps of the race, with $D(m)$ the first walker's lead:

m	$\mathcal{H}_p[m]$	Π	$\sum \Pi(\mathcal{H}_p[i])$	$\mathbb{P}[m+2]$	Π	$\sum \Pi(\mathbb{P}[i])$	$D(m)$
1	5	-1	-1	5	-1	-1	0
2	13	$+1$	0	7	$+1$	0	0
3	17	-1	-1	11	-1	-1	0
4	29	-1	-2	13	$+1$	0	-2
5	37	$+1$	-1	17	-1	-1	0
6	41	-1	-2	19	$+1$	0	-2

The natural question about any race is: how long must one wait for the trailing runner to lead, and by how much? That waiting time is the machinery this section needs:

Definition 8. The *First Deviation* function, Δ_p, is defined by:[4]

$$\Delta_p(n) = \min \left\{ m \;\middle|\; n \leq \sum_{i=1}^{m} \Pi(\mathcal{H}_p[i]) - \sum_{i=3}^{m+2} \Pi(\mathbb{P}[i]) \right\} \tag{9}$$

That is, $\Delta_p(n)$ is the number of steps until the Pythagorean walker first leads by at least n.

Two immediate observations. First, Δ_p is non-decreasing: the set being minimized over shrinks as n grows. Second, since both partial sums are congruent to m modulo 2, the lead $D(m)$ is always *even*; consequently $\Delta_p(2k-1) = \Delta_p(2k)$ for all k. As an entry fact: the Pythagorean walker first noses ahead only at step 41 – that is, $\Delta_p(1) = \Delta_p(2) = 41$.

We first record a fact that we had originally stated as a conjecture, but which follows from the prime number theorem for arithmetic progressions:

[4]If the set on the right-hand side is empty, we set $\Delta_p(n) = \infty$.

Proposition 1. $\lim_{n\to\infty} \frac{\sum_{i=1}^{n} \Pi(\mathcal{H}_p[i])}{n} = 0.$

Proof. Every $p \in \mathcal{H}_p$ satisfies $p \equiv 1 \pmod 4$, and $\Pi(p) = +1$ or -1 according as $p \equiv 1$ or $5 \pmod{12}$. By the prime number theorem for arithmetic progressions [2], the primes are equidistributed among the $\varphi(12) = 4$ reduced residue classes $1, 5, 7, 11 \pmod{12}$; in particular, of the first n Pythagorean primes, $n/2 + o(n)$ lie in each of the classes 1 and $5 \pmod{12}$. The running sum is the difference of these two counts, hence is $o(n)$. □

Thus the ± 1 values $\Pi(\mathcal{H}_p[i])$ average to zero, so $\mathcal{H}_p$ is spread evenly between residues 1 and $-1 \pmod 6$ in the long run, mirroring $\mathbb{P}$ itself. Numerically, however, the cumulative parity of $\mathcal{H}_p$ tends to run *lower* than that of $\mathbb{P}$: over all primes below 2×10^7, the lead $D(m)$ is negative for about 78% of the steps and positive for only 21%. The remaining open conjecture quantifies how often and how far the $\mathcal{H}_p$ sum nonetheless overtakes the $\mathbb{P}$ sum:

Conjecture 1. *The cumulative parity of $\mathcal{H}_p$ exhibits the following non-trivial behavior:*

(a) $\forall n \in \mathbb{N}, \Delta_p(n) < \infty$;

(b) $\lim_{n\to\infty} \frac{\Delta_p(n)}{n^{\ln(2\pi)}} = 41.$[5]

Remark 2. Part (b) implies part (a): Δ_p is non-decreasing, so if $\Delta_p(n_0) = \infty$ for some n_0, then $\Delta_p(n) = \infty$ for all $n \geq n_0$, contradicting the finiteness of the limit in (b). We state the parts separately because (a) may well be provable without any understanding of the growth rate in (b).

Together with Proposition 1, the conjecture suggests that the Pythagorean primes are "as complex as" the full set of primes when measured by their parity behavior: deviations of any size n do occur ($\Delta_p(n) < \infty$), and they grow at the polynomial rate $n^{\ln(2\pi)}$. This "mixing" is what we mean by saying $\mathcal{H}_p$ is non-trivial in a way similar to $\mathbb{P}$.

To return to where we began: the numbers $5, 13, 25, 65, \ldots$ that carry primitive right triangles on their backs are completely classified – which numbers occur (Theorem 2), and in how many ways (Theorem 3). What remains open is subtler. The primes among them, $5, 13, 17, 29, \ldots$, run their ± 1 race against the full sequence of primes, usually trailing; Conjecture 1 asserts that they nonetheless take the lead by every margin eventually – on a schedule that appears to be polynomial. Proving even part (a) would be a satisfying close to the story that $(3, 4, 5)$ begins.

[5] The exponent $\ln(2\pi) \approx 1.838$ and the constant 41 are based on numerical fitting; we do not have a heuristic derivation, and the reader should know how thin the evidence is. Computing with all primes below 2×10^7 (the first 635,170 Pythagorean primes, reaching leads $n \leq 136$), the ratio $\Delta_p(n)/n^{\ln(2\pi)}$ oscillates between roughly 11 and 112 – falling along each plateau of Δ_p and jumping at each new record – with no visible convergence over this range; note also that 41 is exactly $\Delta_p(1)$, so the constant is anchored by the $n = 1$ data point. What the data defensibly supports is polynomial growth with exponent near 1.8; the precise limit and constant should be treated as provisional, subject to revision as more data becomes available.

A A machine-checked Lean 4 proof of Theorem 3, modulo three classical inputs

The proof below has been mechanically checked by Lean 4 (toolchain `leanprover/lean4:v4.29.1`) against Mathlib. The full project lives at `lean/PythHyp/` in this repository; the proof file is `lean/PythHyp/PythHyp/Basic.lean`. The listing below is included directly from that source file, so it cannot drift from what Lean actually checked.

The decomposition is honest: three classical results – explicitly named (F), (G-easy), and (G-uniq) – are stated as `axioms` rather than proved in this file, because their formalization, while certainly possible in Mathlib, is a substantial project of its own. They are classical theorems, not conjectures:

(F) *Fermat's characterization.* A positive integer $z > 1$ is a primitive Pythagorean hypotenuse iff every prime factor of z is $\equiv 1 \pmod 4$. The prime case is in Mathlib as `Nat.Prime.sq_add_sq`; the general case follows from multiplicativity of the norm in $\mathbb{Z}[i]$.

(G-easy) *Multiple factors yield duplicates.* If $z \in \mathcal{H}$ has at least two distinct prime factors, then z admits at least two distinct unordered primitive representations $a^2 + b^2 = z^2$. This is the easy half of the classical 2^{k-1} count and follows from a single non-trivial factorization of z in $\mathbb{Z}[i]$.

(G-uniq) *Prime powers are unique.* If $z = p^n$ for a Pythagorean prime p and $n \geq 1$, then z admits only one unordered primitive representation. This is the harder half of the Gauss count.

Dirichlet's theorem on primes $\equiv 1 \pmod 4$ is *not* an axiom in this proof: it is invoked directly as the Mathlib lemma `Nat.infinite_setOf_prime_modEq_one`.

Modulo the three classical inputs (F), (G-easy), (G-uniq), every step of Theorem 3 – parts (a), (b), and (c) – is fully formalized, with no `sorry`. Lean reports a clean build, and `#print axioms PrimPythHyp.H_partition` confirms the accounting: the final theorem depends on exactly the three axioms above, together with Lean's standard `propext`, `Classical.choice`, and `Quot.sound`.

```lean
import Mathlib.NumberTheory.PythagoreanTriples
import Mathlib.NumberTheory.SumTwoSquares
import Mathlib.NumberTheory.Zsqrtd.GaussianInt
import Mathlib.NumberTheory.PrimesCongruentOne
import Mathlib.Data.Nat.PrimeFin

open Set

namespace PrimPythHyp

/-- z is a primitive Pythagorean hypotenuse. -/
def IsHyp (z : ℕ) : Prop :=
  ∃ a b : ℕ, 0 < a ∧ 0 < b ∧ Nat.gcd a b = 1 ∧ a^2 + b^2 = z^2

def H : Set ℕ := { z | IsHyp z }
def Hp : Set ℕ := { z | z ∈ H ∧ z.Prime }
```

```lean
/-- Hypotenuses with at least two distinct unordered primitive representations. -/
def Hd : Set ℕ :=
  { z | ∃ a₁ b₁ a₂ b₂ : ℕ,
      0 < a₁ ∧ 0 < b₁ ∧ Nat.gcd a₁ b₁ = 1 ∧ a₁^2 + b₁^2 = z^2 ∧
      0 < a₂ ∧ 0 < b₂ ∧ Nat.gcd a₂ b₂ = 1 ∧ a₂^2 + b₂^2 = z^2 ∧
      ({a₁, b₁} : Finset ℕ) ≠ {a₂, b₂} }

/-- Powers of Pythagorean primes. -/
def Hu : Set ℕ := { z | ∃ p ∈ Hp, ∃ n : ℕ, 0 < n ∧ z = p^n }

---------------------------------------------------------------------
-- Classical building blocks (axiomatized; these are real theorems)
---------------------------------------------------------------------

/-- (F) Fermat's two-squares + multiplicativity in ℤ[i]. -/
axiom mem_H_iff_factors {z : ℕ} (hz : 1 < z) :
    z ∈ H ↔ ∀ p ∈ z.primeFactors, p % 4 = 1

/-- (G-easy) If z has ≥ 2 distinct prime factors (all ≡ 1 mod 4), then z has at least
    two distinct unordered primitive representations a^2+b^2 = z^2.  This is the
    part of the Gauss count we actually use for the partition theorem. -/
axiom multiple_factors_yields_Hd {z : ℕ} (hz : z ∈ H) (hk : 2 ≤ z.primeFactors.card) :
    z ∈ Hd

/-- (G-uniq) If z = p^n for a Pythagorean prime p and n ≥ 1, then z has only one
    unordered primitive representation a^2+b^2 = z^2 (so z ∉ Hd). -/
axiom prime_power_not_in_Hd {p : ℕ} (hp : p ∈ Hp) {n : ℕ} (hn : 0 < n) :
    p^n ∉ Hd

---------------------------------------------------------------------
-- Helpers
---------------------------------------------------------------------

lemma one_lt_of_in_H {z : ℕ} (hz : z ∈ H) : 1 < z := by
  obtain ⟨a, b, ha, hb, _, hab⟩ := hz
  have h1a : 1 ≤ a^2 := Nat.one_le_iff_ne_zero.mpr (pow_ne_zero 2 ha.ne')
  have h1b : 1 ≤ b^2 := Nat.one_le_iff_ne_zero.mpr (pow_ne_zero 2 hb.ne')
  have h2 : 2 ≤ z^2 := by rw [← hab]; omega
  -- From z^2 ≥ 2 we get z ≥ 2.
  rcases Nat.lt_or_ge z 2 with hlt | hge
  · interval_cases z <;> simp_all
  · omega

---------------------------------------------------------------------
-- Theorem 3
---------------------------------------------------------------------

/-- (a) H = Hd ∪ Hu. -/
theorem H_eq_Hd_union_Hu : H = Hd ∪ Hu := by
  ext z
  refine ⟨fun hz => ?_, ?_⟩
  · -- z ∈ H. Split on z.primeFactors.card.
    have hz1 : 1 < z := one_lt_of_in_H hz
    have hz_ne : z ≠ 0 := by omega
    have hcard_pos : 1 ≤ z.primeFactors.card := by
      have : z.primeFactors.Nonempty := by
        rw [Nat.nonempty_primeFactors]; exact hz1
      exact Finset.card_pos.mpr this
    by_cases hk : z.primeFactors.card = 1
    · -- One prime factor: z = p^n for that prime p.
```

```
      right
      obtain ⟨p, hp_eq⟩ := Finset.card_eq_one.mp hk
      have hp_in : p ∈ z.primeFactors := by rw [hp_eq]; exact Finset.mem_singleton_self p
      have hp_prime : p.Prime := Nat.prime_of_mem_primeFactors hp_in
      have hp_dvd : p ∣ z := Nat.dvd_of_mem_primeFactors hp_in
      -- The exponent n = z.factorization p, with z = p^n.
      set n := z.factorization p
      have hn_pos : 0 < n := hp_prime.factorization_pos_of_dvd hz_ne hp_dvd
      have hz_eq : z = p ^ n := by
        -- z = ∏ q in z.primeFactors, q ^ z.factorization q.
        -- Since the only factor is p:
        have hprod : z.factorization.prod (· ^ ·) = z :=
          Nat.prod_factorization_pow_eq_self hz_ne
        have hsupp : z.factorization.support = {p} := by
          rw [Nat.support_factorization, hp_eq]
        rw [← hprod, Finsupp.prod, hsupp, Finset.prod_singleton]
      refine ⟨p, ⟨?_, hp_prime⟩, n, hn_pos, hz_eq⟩
      -- Need p ∈ H. Use mem_H_iff_factors with the fact that p.primeFactors = {p}.
      rw [mem_H_iff_factors hp_prime.one_lt]
      intro q hq
      rw [hp_prime.primeFactors, Finset.mem_singleton] at hq
      rw [hq]
      exact (mem_H_iff_factors hz1).mp hz p hp_in
    · -- ≥ 2 prime factors: z ∈ Hd by multiple_factors_yields_Hd.
      left
      have hk2 : 2 ≤ z.primeFactors.card := by omega
      exact multiple_factors_yields_Hd hz hk2
  · rintro (hd | hu)
    · -- Hd ⊆ H: take the first triple as witness.
      obtain ⟨a, b, _, _, ha, hb, hgcd, heq, _⟩ := hd
      exact ⟨a, b, ha, hb, hgcd, heq⟩
    · -- Hu ⊆ H: powers of Pythagorean primes are in H.
      obtain ⟨p, hpH, n, hn, rfl⟩ := hu
      have hp_prime : p.Prime := hpH.2
      have hpn_gt1 : 1 < p^n := one_lt_pow₀ hp_prime.one_lt hn.ne'
      rw [mem_H_iff_factors hpn_gt1]
      intro q hq
      have hpf : (p^n).primeFactors = {p} := Nat.primeFactors_prime_pow hn.ne' hp_prime
      rw [hpf, Finset.mem_singleton] at hq
      rw [hq]
      exact (mem_H_iff_factors hp_prime.one_lt).mp hpH.1 p hp_prime.mem_primeFactors_self

/-- (b) Hd ∩ Hu = ∅. -/
theorem Hd_inter_Hu : Hd ∩ Hu = ∅ := by
  ext z
  simp only [mem_inter_iff, mem_empty_iff_false, iff_false, not_and]
  rintro hd ⟨p, hpH, n, hn, rfl⟩
  exact prime_power_not_in_Hd hpH hn hd

/-- (c) Hd is infinite.

Strategy: from Dirichlet (`Nat.infinite_setOf_prime_modEq_one` with k = 4),
pick infinitely many primes p ≡ 1 (mod 4) with p ≠ 5. The map p ↦ 5*p sends
each such p into Hd: since 5 and p are distinct Pythagorean primes, 5*p has
two distinct prime factors. The map is injective, so the image is infinite. -/
theorem Hd_infinite : Hd.Infinite := by
  -- The set of primes ≡ 1 (mod 4), with 5 removed.
  set S : Set ℕ := {p | p.Prime ∧ p ≡ 1 [MOD 4]} \ {5}
  have h_pyth : Set.Infinite {p : ℕ | p.Prime ∧ p ≡ 1 [MOD 4]} :=
    Nat.infinite_setOf_prime_modEq_one (by norm_num)
```

```
  have hS_inf : S.Infinite := h_pyth.diff (Set.finite_singleton _)
  -- The map p ↦ 5 * p is injective and lands in Hd.
  apply Set.infinite_of_injOn_mapsTo
    (s := S) (t := Hd) (f := fun p => 5 * p)
  · -- InjOn
    intro p₁ _ p₂ _ h
    dsimp only at h
    omega
  · -- MapsTo
    intro p hpS
    obtain ⟨⟨hp_prime, hp_mod⟩, hp_ne⟩ := hpS
    have hp_ne_5 : p ≠ 5 := fun h => hp_ne (by simp [h])
    have h5_prime : Nat.Prime 5 := by decide
    have hp_pos : 0 < p := hp_prime.pos
    have hp_ne0 : p ≠ 0 := hp_pos.ne'
    have h5p_gt1 : 1 < 5 * p := by
      have : 1 < p := hp_prime.one_lt
      omega
    -- Membership in H by Fermat's characterization.
    have h5_in : (5 : ℕ) ∈ (5*p).primeFactors :=
      Nat.mem_primeFactors.mpr ⟨h5_prime, ⟨p, rfl⟩, by omega⟩
    have hp_in : p ∈ (5*p).primeFactors :=
      Nat.mem_primeFactors.mpr ⟨hp_prime, ⟨5, by ring⟩, by omega⟩
    have h5p_in_H : 5 * p ∈ H := by
      rw [mem_H_iff_factors h5p_gt1]
      rw [Nat.primeFactors_mul (by norm_num) hp_ne0,
        h5_prime.primeFactors, hp_prime.primeFactors]
      intro q hq
      rw [Finset.mem_union, Finset.mem_singleton, Finset.mem_singleton] at hq
      rcases hq with rfl | rfl
      · decide
      · exact hp_mod
    -- (5*p).primeFactors.card ≥ 2.
    have hcard : 2 ≤ (5*p).primeFactors.card := by
      have hsub : ({5, p} : Finset ℕ) ⊆ (5*p).primeFactors := by
        intro x hx
        rw [Finset.mem_insert, Finset.mem_singleton] at hx
        rcases hx with rfl | rfl <;> assumption
      have h2 : ({5, p} : Finset ℕ).card = 2 := by
        rw [Finset.card_insert_of_notMem (by simp [Ne.symm hp_ne_5]),
          Finset.card_singleton]
      calc 2 = _ := h2.symm
        _ ≤ _ := Finset.card_le_card hsub
    exact multiple_factors_yields_Hd h5p_in_H hcard
  · exact hS_inf

/-- Theorem 3 of the paper: the partition theorem. -/
theorem H_partition :
    H = Hd ∪ Hu ∧ Hd ∩ Hu = ∅ ∧ Hd.Infinite :=
  ⟨H_eq_Hd_union_Hu, Hd_inter_Hu, Hd_infinite⟩

end PrimPythHyp
```

Once (F), (G-easy), (G-uniq) are themselves formalized, replacing the `axiom` declarations with `theorem` proofs, this file would become a self-contained, machine-checked proof of Theorem 3 with no classical inputs assumed beyond what is already in Mathlib.

References

[1] G. H. Hardy and E. M. Wright, *An Introduction to the Theory of Numbers*, 6th ed., Oxford University Press, 2008. (Fermat's two-square theorem, §§20.1–20.3; Pythagorean triples, §13.2.)

[2] H. Davenport, *Multiplicative Number Theory*, 3rd ed., Graduate Texts in Mathematics 74, Springer, 2000. (Dirichlet's theorem and the prime number theorem for arithmetic progressions.)

[3] W. Sierpiński, *Pythagorean Triangles*, Dover Publications, 2003. (The 2^{k-1} count of primitive triples with a given hypotenuse.)

[4] R. S. McIntire, *An Alternative Proof of an Elementary Property of Pythagorean Triples*, companion note in this repository (`tex/prop_pythag_triple.tex`), 2026. (The parity structure of a primitive triple via the inscribed circle.)